Chapter 9
A Third Look At ML

More Pattern-Matching

- Last time we saw pattern-matching in function definitions:
  - ```
    fun f 0 = "zero"
    |   f _ = "non-zero";
    ```
- Pattern-matching occurs in several other kinds of ML expressions:
  - ```
    case n of
      0 => "zero" |
      _ => "non-zero";
    ```

Case expression

像C语言中switch-case

match

called a <rule>

match 由1个或多个rule组成.

Match Syntax

- A *rule* is a piece of ML syntax that looks like this:

  <*rule*> **::=** <*pattern*> **=>** <*expression*>

- A *match* consists of one or more rules separated by a vertical bar, like this:

  <*match*> **::=** <*rule*> | <*rule*> **'|'** <*match*>

- Each rule in a match must have the same type of expression on the right-hand side
- A match is not an expression by itself, but forms a part of several kinds of ML expressions

Case Expressions

```
- case 1+1 of
=     3 => "three" |
=     2 => "two" |
=     _ => "hmm";
val it = "two" : string
```

- The case expression is one important kind of ML expressions that uses a match.
- The syntax is

  <case-expr> **::=** **case** *<expression>* **of** *<match>*

- This is a very powerful case construct—unlike many languages, it does more than just compare with constants

Example

```
case x of
  _::_::c::_ => c |
  _::b::_ => b |
  a::_ => a |
  nil => 0
```

取列表中的第三，二，一个元素

- The value of this expression is the third element of the list **x**, if it has at least three; or the second element if **x** has only two; or the first element if **x** has only one; or 0 if **x** is empty.

Generalizes **`if`**

if, else 是 case 的一种特殊形式

`if` exp_1 **`then`** exp_2 **`else`** exp_3

```
case exp1 of
   true => exp2 |
   false => exp3
```

- The case expression can do everything the conditional expression can do. The two expressions above are equivalent.
- So **`if-then-else`** is really just a special case of **`case`**

Predefined Functions

- When an ML language system starts, there are many predefined variables
- Some variables happen to be bound to functions:

It is a variable whose value is a function of type `char -> int`

```
- ord;
val it = fn : char -> int
- ~;
val it = fn : int -> int
```

把fun当成一种值.

Defining Functions

- We have seen the **`fun`** notation for defining new named functions
- You can also define new names for old functions, using **`val`** just as for other kinds of values:

```
- val x = ~;
val x = fn : int -> int
- x 3;
val it = ~3 : int
```

Binds **x** to the function denoted by the **~** operator

Function Values

- Functions in ML *do not have names*
- Just like other kinds of values, function values may be given one or more names by binding them to variables
- Thus "the function **`f`**" should be "the function currently bound to the name **`f`**"
- The **`fun`** syntax does two separate things:
  - Creates a new function value
  - Binds that function value to a name

Anonymous Functions

- Named function:

```
- fun f x = x + 2;
val f = fn : int -> int
- f 1;
val it = 3 : int
```

匿名函数（没有函数名）。

- Anonymous function:

Simply give the keyword **fn** followed by a match.

```
- fn x => x + 2;
val it = fn : int -> int
- (fn x => x + 2) 1;
val it = 3 : int
```

直接给出映射关系。

The **`fn`** Syntax

- Another use of the match syntax

  <*fun-expr*> **`::=`** **`fn`** <*match*>

- Using **`fn`**, we get an expression whose value is an (anonymous) function
- We can define what **`fun`** does in terms of **`val`** and **`fn`**
- These two definitions have the same effect:
  - **`fun f x = x + 2`**
  - **`val f = fn x => x + 2`**

Using Anonymous Functions

- One simple application: when you need a small function in just one place
- Without **`fn`**:

```
- fun intBefore (a,b) = a < b;
val intBefore = fn : int * int -> bool
- quicksort ([1,4,3,2,5], intBefore);
val it = [1,2,3,4,5] : int list
```

- With **`fn`**:

```
- quicksort ([1,4,3,2,5], fn (a,b) => a<b);
val it = [1,2,3,4,5] : int list
- quicksort ([1,4,3,2,5], fn (a,b) => a>b);
val it = [5,4,3,2,1] : int list
```

The **op** keyword

```
- op *;
val it = fn : int * int -> int
- quicksort ([1,4,3,2,5], op <);
val it = [1,2,3,4,5] : int list
```

The expression's value is the function used by this operator

- Binary operators are special functions
- Sometimes you want to treat them like plain functions: to pass **<**, for example, as an argument of type **`int * int -> bool`**
- The keyword **op** before an operator is to extract function denoted by the operator

Higher-order Functions

- Every function has an *order*:
  - A function that does not take any functions as parameters, and does not return a function value, has *order 1*
  - A function that takes a function as a parameter or returns a function value has *order n+1*, where *n* is the order of its highest-order parameter or returned value
- The **`quicksort`** we just saw is a second-order function

Practice

What is the order of functions with each of the following ML types?

```
int * int -> bool
int list * (int * int -> bool) -> int list
int -> int -> int
(int -> int) * (int -> int) -> (int -> int)
int -> bool -> real -> string
```

Currying 科里化(名字)

- We've seen how to get two parameters into a function by passing a 2-tuple:

```
fun f (a,b) = a + b;
```

- Another way is to write a function that takes the first argument, and returns another function that takes the second argument:

```
fun g a = fn b => a+b;
```

anonymous function

- The general name for this is *currying*

Curried Addition

```
- fun f (a,b) = a+b;
val f = fn : int * int -> int
- fun g a = fn b => a+b;
val g = fn : int -> int -> int
- f(2,3);
val it = 5 : int
- g 2 3;
val it = 5 : int
```



- Remember that function application is left-associative
- So `g 2 3` means `((g 2) 3)`

Advantages

- No tuples: we get to write **`g 2 3`** instead of **`f(2,3)`**
- But the real advantage: we get to specialize functions for particular initial parameters
- In effect, **`add2`** is a specialized version of **`g`**. (1st parameter fixed at 2 while the 2nd open.)

```
- val add2 = g 2;
val add2 = fn : int -> int
- add2 3;
val it = 5 : int
- add2 10;
val it = 12 : int
```

Multiple Curried Parameters

- Currying generalizes to any number of parameters

```
- fun f (a,b,c) = a+b+c;
val f = fn : int * int * int -> int
- fun g a = fn b => fn c => a+b+c;
val g = fn : int -> int -> int -> int
- f (1,2,3);
val it = 6 : int
- g 1 2 3;
val it = 6 : int
```

Notation For Currying

- There is a much simpler notation for currying (on the next slide)
- The long notation we have used so far makes the little intermediate anonymous functions explicit

```
fun g a = fn b => fn c => a+b+c;
```

- But as long as you understand how it works, the simpler notation is much easier to read and write

Easier Notation for Currying

- Instead of writing:

```
fun f a = fn b => a+b;
```

- We can just write:

```
fun f a b = a+b;
```

- This generalizes for any number of curried arguments

```
- fun f a b c d = a+b+c+d;
val f = fn : int -> int -> int -> int -> int
```

Predefined Higher-Order Functions

- We will use three important predefined higher-order functions:
  - **`map`**
  - **`foldr`**
  - **`foldl`**
- Actually, **`foldr`** and **`foldl`** are very similar, as you might guess from the names

The **map** Function

全局操作.

- Used to apply a function to every element of a list, and collect a list of results

```
- map ~ [1,2,3,4];
val it = [~1,~2,~3,~4] : int list
- map (fn x => x+1) [1,2,3,4];
val it = [2,3,4,5] : int list
- map (fn x => x mod 2 = 0) [1,2,3,4];
val it = [false,true,false,true] : bool list
- map (op +) [(1,2),(3,4),(5,6)];
val it = [3,7,11] : int list
```

The `foldr` Function

- Used to combine all the elements of a list
- For example, to add up all the elements of a list **`x`**, we could write **`foldr (op +) 0 x`**
- It takes **a function *f***, **a starting value *c***, and **a list *x* = [x_1, …, x_n]** and computes:

$$f\left(x_1, f\left(x_2, \cdots f\left(x_{n-1}, f\left(x_n, c\right)\right)\cdots\right)\right)$$

- So **`foldr (op +) 0 [1,2,3,4]`** evaluates as 1+(2+(3+(4+0)))=10

Examples

$$f\left(x_1, f\left(x_2, \cdots f\left(x_{n-1}, f\left(x_n, c\right)\right)\cdots\right)\right)$$

```
- foldr (op +) 0 [1,2,3,4];
val it = 10 : int
- foldr (op * ) 1 [1,2,3,4];
val it = 24 : int
- foldr (op ^) "" ["abc","def","ghi"];
val it = "abcdefghi" : string
- foldr (op ::) [5] [1,2,3,4];
val it = [1,2,3,4,5] : int list
```

The `foldl` Function

- Used to combine all the elements of a list
- Same results as **`foldr`** in some cases

```
- foldl (op +) 0 [1,2,3,4];
val it = 10 : int
- foldl (op * ) 1 [1,2,3,4];
val it = 24 : int
```

The `foldl` Function

- To add up all the elements of a list **`x`**, we could write **`foldl (op +) 0 x`**
- It takes **a function *f***, **a starting value *c***, and **a list *x* = [x_1, …, x_n]** and computes:

$$f(x_n, f(x_{n-1}, \cdots f(x_2, f(x_1, c))\cdots))$$

- So **`foldl (op +) 0 [1,2,3,4]`** evaluates as 4+(3+(2+(1+0)))=10
- Remember, **`foldr`** did 1+(2+(3+(4+0)))=10

The **`foldl`** Function

- **`foldl`** starts at the **`l`**eft, **`foldr`** starts at the **`r`**ight
- Difference does not matter when the function is associative and commutative, like **`+`** and **`*`**
- For other operations, it does matter

```
- foldr (op ^) "" ["abc","def","ghi"];
val it = "abcdefghi" : string
- foldl (op ^) "" ["abc","def","ghi"];
val it = "ghidefabc" : string
- foldr (op -) 0 [1,2,3,4];
val it = ~2 : int
- foldl (op -) 0 [1,2,3,4];
val it = 2 : int
```

Questions?